Revised Example 4: Three Ways of Estimating Multivariate Change: in SAS PROC MIXED, as well as in Multivariate MLM ("Multilevel SEM") and Single-Level SEM in Mplus v. 8.1 (complete syntax and output available electronically)

These simulated data are from Hoffman (2015) chapter 9, and include 200 girls measured approximately annually from ages 12–18 (time 0 = age 18) on their risky behavior (the outcome, a sum ranging from 10 to 50) and the extent to which their mothers monitored their activities (the time-varying predictor, a mean ranging from 1 to 5, centered at 3). A time-invariant predictor of the conservativeness of mothers' attitudes about the smoking and drinking (a mean ranging from 1 to 5, centered at 4) was also collected at the age 12 occasion. Here are the individual growth trajectories for risky behavior and monitoring:







Multivariate Multilevel Model

Level 1:

$$y_{tid} = dvR\left[\beta_{0iR} + \beta_{1iR}\left(Age_{tiR} - 18\right) + \beta_{2iR}\left(Age_{tiR} - 18\right)^2 + e_{tiR}\right] +$$

$$dvM\left[\beta_{0iM} + \beta_{1iM}\left(Age_{tiM} - 18\right) + e_{tiM}\right]$$

Level 2:

Risky Intercept: $\beta_{0iR} = \gamma_{00R} + \gamma_{01R}\left(Attitudes12_i - 4\right) + U_{0iR}$

Risky Age: $\beta_{1iR} = \gamma_{10R} + \gamma_{11R}\left(Attitudes12_i - 4\right) + U_{1iR}$

Risky Age^2: $\beta_{2iR} = \gamma_{20R}$

Monitor Intercept: $\beta_{0iM} = \gamma_{00M} + \gamma_{01M}\left(Attitudes12_i - 4\right) + U_{0iM}$

Monitor Age: $\beta_{1iM} = \gamma_{10M} + \gamma_{11M}\left(Attitudes12_i - 4\right) + U_{1iM}$

The best-fitting unconditional longitudinal models included fixed quadratic and random linear effects of age for risky behavior, but a random linear effect of age for monitoring (although the fixed linear age slope was nonsignificant). In addition, mother's attitudes significantly predicted the intercept and linear age slope for risky behavior. Although they did not significantly predict monitoring, I have added them here to illustrate computing indirect effects.

Chapter 9 began with person-mean-centering and baseline-centering of monitoring of a time-varying predictor of risky behavior. Both were shown to be inadequate because they do not properly distinguish the intercept, linear age slope, and residual variance contained in the monitoring predictor, each of which could potentially relate to those of risky behavior. So the purpose of this example is to demonstrate alternative software methods of estimating models of multivariate change so that you can decide what approach (software and syntax combination) will be most optimal for your own data. See chapter 9 for the results from a directed path model very similar to 2c.

Undirected Multivariate Growth Model for Risky Behavior and Monitoring in SAS PROC MIXED, controlling risky behavior for time-invariant attitudes (Model 1):

```
* Stack longitudinal data into multivariate longitudinal;
DATA RiskyStacked2; SET RiskyStacked;
  DV="1risky  "; dvR=1; dvM=0; outcome=risky; OUTPUT;
  DV="2monitor"; dvR=0; dvM=1; outcome=mon3;  OUTPUT;
RUN;

TITLE1 "Multivariate Model at Age 18 = Time 0";
PROC MIXED DATA=work.Chapter9 NOCLPRINT COVTEST IC
     NAMELEN=100 METHOD=ML;
CLASS FamilyID occasion DV;

MODEL outcome = dvR dvM dvR*agec18 dvM*agec18 dvR*agec18*agec18
                dvR*att4 dvR*agec18*att4 dvM*att4 dvM*agec18*att4
                / NOINT SOLUTION DDFM=Satterthwaite;
RANDOM dvR dvM dvR*agec18 dvM*agec18
        / G GCORR TYPE=UN SUBJECT=FamilyID;
REPEATED DV / R RCORR TYPE=UN SUBJECT=occasion*FamilyID;
RUN; TITLE1;
```

Results start here: This is the same model as in SAS…

SAS:

```
           Fit Statistics
-2 Log Likelihood            8783.8
AIC (Smaller is Better)      8827.8
AICC (Smaller is Better)     8828.2
BIC (Smaller is Better)      8900.4
```

MPLUS:

```
Number of Free Parameters                        22

Loglikelihood
          H0 Value                        -4391.885

Information Criteria
          Akaike (AIC)                     8827.771
          Bayesian (BIC)                   8943.144
          Sample-Size Adjusted BIC         8873.258
            (n* = (n + 2) / 24)
```

In Mplus, the same Model 1 as an undirected multivariate MLM:

```
TITLE:   Model 1: Undirected Multivariate Growth Model as MLM
DATA:    FILE = Example4.csv;    ! Syntax in same folder as data
VARIABLE:
! List of variables in data file
  NAMES = PersonID occasion risky age18 att4 agesq mon3;
! Variables to be analyzed in this model
  USEVARIABLE = age18 agesq att4 risky mon3;
  MISSING ARE ALL (-999);    ! Missing data identifier
! MLM options
  CLUSTER = PersonID;        ! Level-2 ID
  BETWEEN = att4;            ! Observed ONLY level-2 predictors
  WITHIN  = age18 agesq;     ! Observed ONLY level-1 predictors

ANALYSIS:   TYPE = TWOLEVEL RANDOM; ESTIMATOR = ML;

MODEL:  ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar);   ! L1 R: Residual variances (labels)
 Rslp  | risky ON age18;         ! Placeholder for R linear age slope
 Rquad | risky ON agesq;         ! Placeholder for R quadratic age slope
 Mslp  | mon3  ON age18;         ! Placeholder for M linear age slope
 risky WITH mon3 (ResCov);       ! L1 R: Residual covariance

%BETWEEN%
[risky mon3];                    ! Fixed intercepts
 risky mon3 (Rintvar Mintvar);   ! L2 G: Random intercept variances (labels)
[Rslp Rquad Mslp];               ! Fixed age slopes (as defined earlier)
 Rslp Mslp  (Rslpvar Mslpvar);   ! L2 G: Random linear age slope variances
 Rquad@0;                        ! No quadratic age slope variance

 risky Rslp ON att4;             ! Att-> R int, linear age slope
  mon3 Mslp ON att4;             ! Att-> M int, linear age slope
 risky WITH Rslp  (RIntSlp);     ! R Int-slope covariance (label)
 mon3  WITH Mslp  (MIntSlp);     ! M Int-slope covariance (label)

 risky WITH mon3  (IntCov);      ! L2 G: Random intercept covariance
 Rslp  WITH Mslp  (SlpCov);      ! L2 G: Random linear age slope covariance
 mon3  WITH Rslp  (Int2Slp);     ! L2 G: M int, R slope covariance
 Mslp  WITH risky (Slp2Int);     ! L2 G: M slope, R int covariance

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each new created effect
NEW(ResCor IntCor SlpCor RIScor MIScor I2SCor S2ICor);

! Estimating correlations found in SAS RCORR and GCORR
! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov  / (SQRT(Rresvar)*SQRT(Mresvar));
  IntCor = IntCov  / (SQRT(Rintvar)*SQRT(Mintvar));
  SlpCor = SlpCov  / (SQRT(Rslpvar)*SQRT(Mslpvar));
  RIScor = RIntSlp / (SQRT(Rintvar)*SQRT(Rslpvar));
  MIScor = MIntSlp / (SQRT(Mintvar)*SQRT(Mslpvar));
  I2Scor = Int2Slp / (SQRT(Mintvar)*SQRT(Rslpvar));
  S2Icor = Slp2Int / (SQRT(Mslpvar)*SQRT(Rintvar));
```

SAS Undirected Multivariate MLM Results:

Estimated R Matrix for PersonID*occasion 1 12

Row	Col1	Col2
1	8.3538	0.2874
2	0.2874	0.08077

Estimated R Correlation Matrix for PersonID*occasion 1 12

Row	Col1	Col2
1	1.0000	0.3499
2	**0.3499**	1.0000

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2	Col3	Col4
1	dvR	1	18.0535	-0.8527	1.8821	0.04051
2	dvM	1	-0.8527	0.1946	-0.1062	-0.00042
3	dvR*agec18	1	1.8821	-0.1062	0.4882	-0.01817
4	dvM*agec18	1	0.04051	-0.00042	-0.01817	0.01048

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2	Col3	Col4
1	dvR	1	1.0000	-0.4549	0.6339	0.0931
2	dvM	1	**-0.4549**	1.0000	-0.3446	-0.0093
3	dvR*agec18	1	**0.6339**	**-0.3446**	1.0000	-0.2539
4	dvM*agec18	1	**0.0931**	**-0.0093**	**-0.2539**	1.0000

Covariance Parameter Estimates (covariances only to save space)

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(2,1)	PersonID	-0.8527	0.1680	-5.08	<.0001
UN(3,1)	PersonID	1.8821	0.3562	5.28	<.0001
UN(3,2)	PersonID	-0.1062	0.03077	-3.45	0.0006
UN(4,1)	PersonID	0.04051	0.03877	1.04	0.2961
UN(4,2)	PersonID	-0.00042	0.004000	-0.11	0.9164
UN(4,3)	PersonID	-0.01817	0.007341	-2.47	0.0133
UN(2,1)	PersonID*occasion	0.2874	0.02753	10.44	<.0001

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
dvR	23.3224	0.3477	239	67.07	<.0001
dvM	0.06287	0.03418	200	1.84	0.0674
dvR*agec18	1.9749	0.1386	1185	14.25	<.0001
dvM*agec18	-0.00312	0.008202	200	-0.38	0.7040
dvR*agec18*agec18	0.1466	0.02058	1010	7.12	<.0001
dvR*att4	-3.1601	0.5509	200	-5.74	<.0001
dvR*agec18*att4	-0.5173	0.1043	199	-4.96	<.0001
dvM*att4	-0.04418	0.05668	200	-0.78	0.4366
dvM*agec18*att4	0.003269	0.01360	200	0.24	0.8103

Mplus results continue: This is the same model as in SAS…

```
                                                    Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value
Within Level
RISKY    WITH
    MON3               0.287      0.028     10.441      0.000
 Residual Variances
    RISKY              8.352      0.374     22.351      0.000
    MON3               0.081      0.004     22.354      0.000

Between Level
 RSLP       ON
    ATT4              -0.518      0.104     -4.963      0.000
 MSLP       ON
    ATT4               0.003      0.014      0.240      0.810
 RISKY      ON
    ATT4              -3.160      0.551     -5.737      0.000
 MON3       ON
    ATT4              -0.044      0.057     -0.779      0.436
 RISKY    WITH
    RSLP               1.878      0.356      5.273      0.000
    MSLP               0.040      0.039      1.044      0.296
 MON3     WITH
    MSLP               0.000      0.004     -0.105      0.916
    RSLP              -0.106      0.031     -3.449      0.001
 RSLP     WITH
    MSLP              -0.018      0.007     -2.478      0.013
 RISKY    WITH
    MON3              -0.853      0.168     -5.076      0.000

 Means
    RQUAD              0.147      0.021      7.117      0.000
 Intercepts
    RISKY             23.322      0.348     67.075      0.000
    MON3               0.063      0.034      1.839      0.066
    RSLP               1.975      0.138     14.259      0.000
    MSLP              -0.003      0.008     -0.380      0.704
  Variances
    RQUAD              0.000      0.000    999.000    999.000
 Residual Variances
    RISKY             18.049      2.202      8.198      0.000
    MON3               0.195      0.023      8.371      0.000
    RSLP               0.484      0.080      6.071      0.000
    MSLP               0.010      0.001      7.802      0.000

New/Additional Parameters – look for the first 3 across outputs
    RESCOR             0.350      0.028     12.607      0.000
    INTCOR            -0.455      0.074     -6.119      0.000
    SLPCOR            -0.255      0.103     -2.483      0.013
    RISCOR             0.635      0.057     11.088      0.000
    MISCOR            -0.009      0.089     -0.105      0.917
    I2SCOR            -0.346      0.095     -3.646      0.000
    S2ICOR             0.093      0.087      1.066      0.286
```

In Mplus, the same Model 1 as an undirected single-level SEM:

```
TITLE:   Model 1: Undirected Multivariate Growth Model as Single-Level SEM
DATA:    FILE = Example4.csv;    ! Syntax in same folder as data
! Unstacking to multivariate format
DATA LONGTOWIDE:
! Names of old stacked former variables (without numbers)
  LONG = risky|mon|age;
! Names of new multivariate variables (that use numbers)
  WIDE = risky12-risky18|mon12-mon18|age12-age18;
! Variable with level-2 ID info
  IDVARIABLE = PersonID;
! Old level-1 identifier
  REPETITION = age (12 13 14 15 16 17 18);
VARIABLE:
! List of variables in original data file
  NAMES = PersonID occasion risky age18 att4 mon3 agesq;
! Variables to be analyzed in this model
  USEVARIABLE = att4 age12-age18 mon12-mon18 risky12-risky18;
  MISSING ARE ALL (-999);   ! Missing data identifier
  TSCORES = age12-age18;    ! Exact time indicator

ANALYSIS:   TYPE = RANDOM; ESTIMATOR = ML; MODEL = NOCOVARIANCES;
MODEL:  ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0
 risky12-risky18 (Rresvar);      ! L1 R: R residual variances held equal
 mon12-mon18     (Mresvar);      ! L1 R: M residual variances held equal

! Risky behavior quadratic growth model using exact age as loadings
  Rint Rslp Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear growth model using exact age as loadings
  Mint Mslp | mon12-mon18 AT age12-age18;
! Fixed growth effects for R and M
  [Rint Rslp Rquad Mint Mslp];
! L2 G: Random int and linear age slope variances, no quad age variance
  Rint Rslp Mint Mslp (Rintvar Rslpvar Mintvar Mslpvar); Rquad@0;
! L2 G: Within-variable random int-slope covariances for R, M
  Rint WITH Rslp (Rintslp); Mint WITH Mslp (Mintslp);
! Attitudes --> R int, R linear slope, M int, M linear slope
  Rint Rslp Mint Mslp ON att4;

! Covariances between outcomes
 Rint WITH Mint (IntCov);      ! L2 G: Random intercept covariance
 Rslp WITH Mslp (SlpCov);      ! L2 G: Random linear age slope covariance
 Mint WITH Rslp (Int2Slp);     ! L2 G: M int, R slope covariance
 Mslp WITH Rint (Slp2Int);     ! L2 G: M slope, R int covariance
! Residual WP covariance between same ages, held equal across age
  risky12-risky18 PWITH mon12-mon18 (ResCov);

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each new created effect
NEW(ResCor IntCor SlpCor);
! Estimating correlations found in SAS RCORR and GCORR
! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov  / (SQRT(Rresvar)*SQRT(Mresvar));
  IntCor = IntCov  / (SQRT(Rintvar)*SQRT(Mintvar));
  SlpCor = SlpCov  / (SQRT(Rslpvar)*SQRT(Mslpvar));
```




Indicates paths fixed = 1

Indicates paths fixed = time values

Indicates paths freely estimated

Indicates paths freely estimated between residuals at the same occasion but held equal over time

For balanced time, a linear growth model would look like this instead (add quad as third place before |):

```
Mint Mslp | mon12@-6 mon13@-5 mon14@-4 mon15@-3
            mon16@-2 mon17@-1 mon18@0;
```

MODEL RESULTS

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
RINT ON				
ATT4	-3.160	0.551	-5.737	0.000
RSLP ON				
ATT4	-0.518	0.104	-4.963	0.000
MINT ON				
ATT4	-0.044	0.057	-0.779	0.436
MSLP ON				
ATT4	0.003	0.014	0.240	0.810
RINT WITH				
RSLP	1.878	0.356	5.273	0.000
MINT	-0.853	0.168	-5.076	0.000
MSLP	0.040	0.039	1.044	0.296
MINT WITH				
MSLP	0.000	0.004	-0.105	0.916
RSLP	-0.106	0.031	-3.449	0.001
RSLP WITH				
MSLP	-0.018	0.007	-2.478	0.013
RISKY12 WITH				
MON12	0.287	0.028	10.441	0.000
RISKY13 WITH				
MON13	0.287	0.028	10.441	0.000
RISKY14 WITH				
MON14	0.287	0.028	10.441	0.000
RISKY15 WITH				
MON15	0.287	0.028	10.441	0.000
RISKY16 WITH				
MON16	0.287	0.028	10.441	0.000
RISKY17 WITH				
MON17	0.287	0.028	10.441	0.000
RISKY18 WITH				
MON18	0.287	0.028	10.441	0.000
Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
RISKY12	0.000	0.000	999.000	999.000
RISKY13	0.000	0.000	999.000	999.000
RISKY14	0.000	0.000	999.000	999.000
RISKY15	0.000	0.000	999.000	999.000
RISKY16	0.000	0.000	999.000	999.000
RISKY17	0.000	0.000	999.000	999.000
RISKY18	0.000	0.000	999.000	999.000
MON12	0.000	0.000	999.000	999.000
MON13	0.000	0.000	999.000	999.000
MON14	0.000	0.000	999.000	999.000
MON15	0.000	0.000	999.000	999.000
MON16	0.000	0.000	999.000	999.000
MON17	0.000	0.000	999.000	999.000
MON18	0.000	0.000	999.000	999.000
RINT	23.322	0.348	67.074	0.000
RSLP	1.975	0.138	14.259	0.000
MINT	0.063	0.034	1.839	0.066
MSLP	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
RISKY12	8.352	0.374	22.351	0.000
RISKY13	8.352	0.374	22.351	0.000
RISKY14	8.352	0.374	22.351	0.000
RISKY15	8.352	0.374	22.351	0.000
RISKY16	8.352	0.374	22.351	0.000
RISKY17	8.352	0.374	22.351	0.000
RISKY18	8.352	0.374	22.351	0.000
MON12	0.081	0.004	22.354	0.000
MON13	0.081	0.004	22.354	0.000
MON14	0.081	0.004	22.354	0.000
MON15	0.081	0.004	22.354	0.000
MON16	0.081	0.004	22.354	0.000
MON17	0.081	0.004	22.354	0.000
MON18	0.081	0.004	22.354	0.000
RINT	18.049	2.202	8.198	0.000
RSLP	0.484	0.080	6.071	0.000
MINT	0.195	0.023	8.371	0.000
MSLP	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESCOR	**0.350**	**0.028**	**12.607**	**0.000**
INTCOR	**-0.455**	**0.074**	**-6.119**	**0.000**
SLPCOR	**-0.255**	**0.103**	**-2.483**	**0.013**

Model 2a: Partially Directed Path Multivariate MLM in Mplus: Monitor → Risky for intercepts and slopes, but residuals covary

```
TITLE:  Model 2a: Partially Directed Multivariate Growth Model as MLM
        L1 WP effect as residual covariance

( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 )

MODEL:  ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar);  ! L1 R: Residual variances (labels)
 Rslp  | risky ON age18;        ! Placeholder for R linear age slope
 Rquad | risky ON agesq;        ! Placeholder for R quadratic age slope
 Mslp  | mon3  ON age18;        ! Placeholder for M linear age slope
 risky WITH mon3 (ResCov);      ! L1 R: Still residual covariance

%BETWEEN%
[risky mon3];                   ! Fixed intercepts
 risky mon3 (Rintvar Mintvar);  ! L2 G: Random intercept variances (labels)
[Rquad Rslp Mslp];              ! Fixed age slopes (as defined earlier)
 Rslp Mslp  (Rslpvar Mslpvar);  ! L2 G: Random linear age slope variances
 Rquad@0;                       ! No quadratic age slope variance
 risky Rslp ON att4;            ! Att-> R int, linear age slope
  mon3 Mslp ON att4;            ! Att-> M int, linear age slope
 risky WITH Rslp (RIntSlp);     ! R Int-slope covariance (label)
 mon3  WITH Mslp (MIntSlp);     ! M Int-slope covariance (label)

! Although we have changed the int-int and slope-slope relations to direct
! paths from M -> R instead of covariances, they still represent total BP
! relationships because the L1 relationship is still a covariance

 risky ON   mon3 (BPIntEff);    ! Total BP intercept effect
 Rslp  ON   Mslp (BPSlpEff);    ! Total BP age slope effect

 mon3  WITH Rslp  (Int2Slp);    ! L2 G: M int, R slope covariance
 Mslp  WITH risky (Slp2Int);    ! L2 G: M slope, R int covariance

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each new created effect -- all values from undirected model
NEW(ResCor IntStd SlpStd);

! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov / (SQRT(8.3538)*SQRT(0.08077));      ! WP Res corr

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644);   ! STD BP Int effect
  SlpStd = BPSlpEff * SQRT(0.01049) / SQRT(0.48830);   ! STD BP Slope effect
```

This is the same model, just with a different way of specifying the level-2 intercept to intercept and slope to slope relationships.

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
RISKY WITH				
MON3	0.287	0.028	10.441	0.000
Residual Variances				
RISKY	8.352	0.374	22.351	0.000
MON3	0.081	0.004	22.354	0.000
Between Level – **new parameters are in BOLD underline**				
RSLP ON				
MSLP	**-1.736**	**0.713**	**-2.434**	**0.015**
RSLP ON				
ATT4	-0.512	0.105	-4.869	0.000
MSLP ON				
ATT4	0.003	0.014	0.240	0.810
RISKY ON				
ATT4	-3.354	0.528	-6.348	0.000
MON3	**-4.380**	**0.797**	**-5.497**	**0.000**
MON3 ON				
ATT4	-0.044	0.057	-0.779	0.436
RISKY WITH				
RSLP	1.480	0.345	4.285	0.000
MSLP	0.039	0.038	1.023	0.306
MON3 WITH				
MSLP	0.000	0.004	-0.105	0.916
RSLP	-0.107	0.031	-3.454	0.001
Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
RISKY	23.598	0.338	69.837	0.000
MON3	0.063	0.034	1.839	0.066
RSLP	1.969	0.139	14.195	0.000
MSLP	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
RISKY	14.315	2.030	7.053	0.000
MON3	0.195	0.023	8.371	0.000
RSLP	0.453	0.081	5.564	0.000
MSLP	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESCOR	**0.350**	**0.034**	**10.441**	**0.000**
INTSTD	**-0.455**	**0.083**	**-5.497**	**0.000**
SLPSTD	**-0.254**	**0.104**	**-2.434**	**0.015**

In Mplus, Model 2a as a partially directed single-level SEM: Monitor → Risky for intercepts and slopes, but residuals covary

```
TITLE:  Model 2a: Partially Directed Multivariate Growth Model as
        Single-Level SEM, L1 WP effect as residual covariance

( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 )

MODEL:  ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0
 risky12-risky18 (Rresvar);      ! L1 R: R residual variances held equal
 mon12-mon18     (Mresvar);      ! L1 R: M residual variances held equal

! Risky behavior quadratic growth model using exact age as loadings
  Rint Rslp Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear growth model using exact age as loadings
  Mint Mslp | mon12-mon18 AT age12-age18;
! Fixed growth effects for R and M
  [Rint Rslp Rquad Mint Mslp];
! L2 G: Random int and linear age slope variances, no quad age variance
  Rint Rslp Mint Mslp (Rintvar Rslpvar Mintvar Mslpvar); Rquad@0;
! L2 G: Within-variable random int-slope covariances for R, M
  Rint WITH Rslp (Rintslp); Mint WITH Mslp (Mintslp);
! Attitudes --> R int, R linear slope, M int, M linear slope
  Rint Rslp Mint Mslp ON att4;

! Although we have changed the int-int and slope-slope relations to direct
! paths from M -> R instead of covariances, they still represent total BP
! relationships because the L1 relationship is still a covariance

  Rint ON Mint  (BPIntEff);     ! Total BP intercept effect
  Rslp ON Mslp  (BPSlpEff);     ! Total BP age slope effect

  Mint WITH Rslp (Int2Slp);     ! L2 G: M int, R slope covariance
  Mslp WITH Rint (Slp2Int);     ! L2 G: M slope, R int covariance

! Residual WP covariance between same ages, held equal across age
  risky12-risky18 PWITH mon12-mon18 (ResCov);

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each new created effect -- all values from undirected model
  NEW(ResCor IntStd SlpStd);

! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov / (SQRT(8.3538)*SQRT(0.08077));    ! WP Res corr

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644); ! STD BP Int effect
  SlpStd = BPSlpEff * SQRT(0.01049) / SQRT(0.48830); ! STD BP Slope effect
```

This is the same model, just with a different way of specifying the level-2 intercept to intercept and slope to slope relationships.




```
For balanced time, a linear growth model would look like this
instead (add quad as third place before |):

Mint Mslp | mon12@-6 mon13@-5 mon14@-4 mon15@-3
            mon16@-2 mon17@-1 mon18@0;
```

MODEL RESULTS – **changed parameters are in BOLD underline**

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
RINT ON				
MINT	**-4.380**	**0.797**	**-5.496**	**0.000**
RSLP ON				
MSLP	**-1.736**	**0.713**	**-2.434**	**0.015**
RINT ON				
ATT4	-3.354	0.528	-6.348	0.000
RSLP ON				
ATT4	-0.512	0.105	-4.869	0.000
MINT ON				
ATT4	-0.044	0.057	-0.779	0.436
MSLP ON				
ATT4	0.003	0.014	0.240	0.810
RINT WITH				
RSLP	1.480	0.345	4.285	0.000
MSLP	0.039	0.038	1.023	0.306
MINT WITH				
MSLP	0.000	0.004	-0.105	0.916
RSLP	-0.107	0.031	-3.454	0.001
RISKY12 WITH				
MON12	0.287	0.028	10.441	0.000
RISKY13 WITH				
MON13	0.287	0.028	10.441	0.000
RISKY14 WITH				
MON14	0.287	0.028	10.441	0.000
RISKY15 WITH				
MON15	0.287	0.028	10.441	0.000
RISKY16 WITH				
MON16	0.287	0.028	10.441	0.000
RISKY17 WITH				
MON17	0.287	0.028	10.441	0.000
RISKY18 WITH				
MON18	0.287	0.028	10.441	0.000

Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
RISKY12	0.000	0.000	999.000	999.000
RISKY13	0.000	0.000	999.000	999.000
RISKY14	0.000	0.000	999.000	999.000
RISKY15	0.000	0.000	999.000	999.000
RISKY16	0.000	0.000	999.000	999.000
RISKY17	0.000	0.000	999.000	999.000
RISKY18	0.000	0.000	999.000	999.000
MON12	0.000	0.000	999.000	999.000
MON13	0.000	0.000	999.000	999.000
MON14	0.000	0.000	999.000	999.000
MON15	0.000	0.000	999.000	999.000
MON16	0.000	0.000	999.000	999.000
MON17	0.000	0.000	999.000	999.000
MON18	0.000	0.000	999.000	999.000
RINT	23.598	0.338	69.836	0.000
RSLP	1.969	0.139	14.195	0.000
MINT	0.063	0.034	1.839	0.066
MSLP	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
RISKY12	8.352	0.374	22.351	0.000
RISKY13	8.352	0.374	22.351	0.000
RISKY14	8.352	0.374	22.351	0.000
RISKY15	8.352	0.374	22.351	0.000
RISKY16	8.352	0.374	22.351	0.000
RISKY17	8.352	0.374	22.351	0.000
RISKY18	8.352	0.374	22.351	0.000
MON12	0.081	0.004	22.354	0.000
MON13	0.081	0.004	22.354	0.000
MON14	0.081	0.004	22.354	0.000
MON15	0.081	0.004	22.354	0.000
MON16	0.081	0.004	22.354	0.000
MON17	0.081	0.004	22.354	0.000
MON18	0.081	0.004	22.354	0.000
RINT	14.315	2.030	7.053	0.000
RSLP	0.453	0.081	5.564	0.000
MINT	0.195	0.023	8.371	0.000
MSLP	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESCOR	**0.350**	**0.034**	**10.441**	**0.000**
INTSTD	**-0.455**	**0.083**	**-5.496**	**0.000**
SLPSTD	**-0.254**	**0.104**	**-2.434**	**0.015**

Model 2b: Partially Directed Path Multivariate MLM in Mplus: Monitor → Risky for WP residuals within L1 model
Also demonstrating how to request BP indirect effects

```
TITLE:  Model 2b: Partially Directed Multivariate Growth Model as MLM
        L1 WP effect as direct path specified in L1 WITHIN

( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 )

MODEL:  ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar);  ! L1 R: Residual variances (labels)
 Rslp  | risky ON age18;        ! Placeholder for R linear age slope
 Rquad | risky ON agesq;        ! Placeholder for R quadratic age slope
 Mslp  | mon3  ON age18;        ! Placeholder for M linear age slope
 risky ON mon3 (ResEff);        ! L1 WP fixed effect M->R here (label)

%BETWEEN%
[risky mon3];                   ! Fixed intercepts
 risky mon3 (Rintvar Mintvar);  ! L2 G: Random intercept variances (labels)
[Rquad Rslp Mslp];              ! Fixed age slopes (as defined earlier)
 Rslp Mslp  (Rslpvar Mslpvar);  ! L2 G: Random linear age slope variances
 Rquad@0;                       ! No quadratic age slope variance
 risky Rslp ON att4 (XtoYint XtoYslp);  ! Att-> R int, linear age slope
  mon3 Mslp ON att4 (XtoMint XtoMslp);  ! Att-> M int, linear age slope
 risky WITH Rslp (RIntSlp);     ! R Int-slope covariance (label)
  mon3 WITH Mslp (MIntSlp);     ! M Int-slope covariance (label)

! Although the intercept -> intercept path remains the total BP effect,
! now the slope -> slope path becomes the contextual BP effect instead

 risky ON   mon3  (BPIntEff);   ! STILL total BP intercept effect
 Rslp  ON   Mslp  (SlpCont);    ! NOW contextual BP age slope effect

 mon3  WITH Rslp  (Int2Slp);    ! L2 G: M int, R slope covariance
 Mslp  WITH risky (Slp2Int);    ! L2 G: M slope, R int covariance

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each new created effect -- all values from undirected model
NEW(ResStd IntStd BPSlpEff SlpStd indBPint indBPslp);

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStd = ResEff   * SQRT(0.08077) / SQRT(8.3538);  ! STD WP Res effect
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644); ! STD BP Int effect
  BPSlpEff = ResEff + SlpCont;                       ! WP + Context = BP slp
  SlpStd = BPSlpEff * SQRT(0.01049) / SQRT(0.48830); ! STD BP Slope effect
  indBPint = XtoMint * BPIntEff;             ! BP intercept indirect effect
  indBPslp = XtoMslp * BPSlpEff;             ! BP age slope indirect effect
```

This is still the same model, just with a different way of specifying the level-1 residual to residual relationship. This method will only work for level-1 effects that are fixed, though.

```
                                                    Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value

Within Level – new parameters are in BOLD underline
 RISKY      ON
    MON3               3.559      0.301     11.809      0.000
 Residual Variances
    RISKY              7.329      0.328     22.353      0.000
    MON3               0.081      0.004     22.354      0.000

Between Level – changed parameters are in BOLD underline
 RSLP       ON
    MSLP              -5.294      0.806     -6.569      0.000
 RSLP       ON
    ATT4              -0.512      0.105     -4.869      0.000
 MSLP       ON
    ATT4               0.003      0.014      0.240      0.810
 RISKY      ON
    ATT4              -3.354      0.528     -6.348      0.000
    MON3              -4.380      0.797     -5.497      0.000
 MON3       ON
    ATT4              -0.044      0.057     -0.779      0.436

 RISKY    WITH
    RSLP               1.480      0.345      4.285      0.000
    MSLP               0.039      0.038      1.023      0.306
 MON3     WITH
    MSLP               0.000      0.004     -0.105      0.916
    RSLP              -0.107      0.031     -3.454      0.001
 Means
    RQUAD              0.147      0.021      7.117      0.000
 Intercepts
    RISKY             23.598      0.338     69.837      0.000
    MON3               0.063      0.034      1.839      0.066
    RSLP               1.969      0.139     14.195      0.000
    MSLP              -0.003      0.008     -0.380      0.704
 Variances
    RQUAD              0.000      0.000    999.000    999.000
 Residual Variances
    RISKY             14.315      2.030      7.053      0.000
    MON3               0.195      0.023      8.371      0.000
    RSLP               0.453      0.081      5.564      0.000
    MSLP               0.010      0.001      7.802      0.000

New/Additional Parameters
    RESCOR             0.350      0.034     10.441      0.000
    INTSTD            -0.455      0.083     -5.497      0.000
    BPSLPEFF          -1.736      0.713     -2.434      0.015
    SLPSTD            -0.254      0.104     -2.434      0.015

    INDBPINT           0.194      0.251      0.772      0.440
    INDBPSLP          -0.006      0.024     -0.239      0.811
```

In Mplus, Model 2b as a partially directed single-level SEM:
Monitor → Risky for WP residuals using structured residuals

```
TITLE:  Model 2b: Partially Directed Multivariate Growth Model as Single-Level
        SEM, L1 WP effect as direct path using structured residuals
( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 )
MODEL:  ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0

! Risky behavior quadratic growth model using exact age as loadings
  Rint Rslp Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear growth model using exact age as loadings
  Mint Mslp | mon12-mon18 AT age12-age18;
! Fixed growth effects for R and M
  [Rint Rslp Rquad Mint Mslp];
! L2 G: Random int and linear age slope variances, no quad age variance
  Rint Rslp Mint Mslp (Rintvar Rslpvar Mintvar Mslpvar); Rquad@0;
! L2 G: Within-variable random int-slope covariances for R, M
  Rint WITH Rslp (Rintslp); Mint WITH Mslp (Mintslp);
! Attitudes --> R int, R linear slope, M int, M linear slope
  Rint Rslp Mint Mslp ON att4 (XtoYint XtoYslp XtoMint XtoMslp);
  Rint ON Mint  (BPIntEff);    ! Total BP intercept effect
  Rslp ON Mslp  (BPSlpEff);    ! Total BP age slope effect
  Mint WITH Rslp (Int2Slp);    ! L2 G: M int, R slope covariance
  Mslp WITH Rint (Slp2Int);    ! L2 G: M slope, R int covariance

! Define new latent factors for residuals at each occasion
  Frisky12 BY risky12@1; Frisky13 BY risky13@1; Frisky14 BY risky14@1;
  Frisky15 BY risky15@1; Frisky16 BY risky16@1; Frisky17 BY risky17@1;
  Frisky18 BY risky18@1; Fmon12 BY mon12@1; Fmon13 BY mon13@1;
  Fmon14 BY mon14@1; Fmon15 BY mon15@1; Fmon16 BY mon16@1;
  Fmon17 BY mon17@1; Fmon18 BY mon18@1;
! All factor means fixed to 0
  [Frisky12-Frisky18@0 Fmon12-Fmon18@0];
! Shut off old residual variances
  risky12-risky18@0 mon12-mon18@0;
! Hold new residual variances equal over time
  Frisky12-Frisky18 (Rresvar); ! L1 R: R residual variances held equal
  Fmon12-Fmon18     (Mresvar); ! L1 R: M residual variances held equal
! Factor residual WP effect between same ages, held equal across age
  Frisky12-Frisky18 PON Fmon12-Fmon18 (ResEff);

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each new created effect -- all values from undirected model
NEW(ResStd IntStd SlpStd indBPint indBPslp);
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStd = ResEff   * SQRT(0.08077) / SQRT(8.3538);  ! STD WP Res effect
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644); ! STD BP Int effect
  SlpStd = BPSlpEff * SQRT(0.01049) / SQRT(0.48830); ! STD BP Slope effect
  indBPint = XtoMint * BPIntEff;           ! BP intercept indirect effect
  indBPslp = XtoMslp * BPSlpEff;           ! BP age slope indirect effect
```

This is the same model, just with a different way of specifying the level-2 intercept to intercept and slope to slope relationships.



For balanced time, a linear growth model would look like this instead (add quad as third place before |):

```
Mint Mslp | mon12@-6 mon13@-5 mon14@-4 mon15@-3
            mon16@-2 mon17@-1 mon18@0;
```

MODEL RESULTS – **changed parameters are in BOLD underline**

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Factor loadings set to 1 omitted				
RINT ON				
MINT	-4.380	0.797	-5.496	0.000
RSLP ON				
MSLP	-1.736	0.713	-2.434	0.015
FRISKY12 ON				
FMON12	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY13 ON				
FMON13	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY14 ON				
FMON14	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY15 ON				
FMON15	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY16 ON				
FMON16	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY17 ON				
FMON17	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY18 ON				
FMON18	**3.563**	**0.302**	**11.810**	**0.000**
RINT ON				
ATT4	-3.354	0.528	-6.348	0.000
RSLP ON				
ATT4	-0.512	0.105	-4.869	0.000
MINT ON				
ATT4	-0.044	0.057	-0.779	0.436
MSLP ON				
ATT4	0.003	0.014	0.240	0.810
RINT WITH				
RSLP	1.480	0.345	4.285	0.000
MSLP	0.039	0.038	1.023	0.306
MINT WITH				
MSLP	0.000	0.004	-0.105	0.916
RSLP	-0.107	0.031	-3.454	0.001
Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
Intercepts fixed to 0 omitted				
RINT	23.598	0.338	69.836	0.000
RSLP	1.969	0.139	14.195	0.000
MINT	0.063	0.034	1.839	0.066
MSLP	-0.003	0.008	-0.380	0.704
Variances				
FMON12	0.081	0.004	22.327	0.000
FMON13	0.081	0.004	22.327	0.000
FMON14	0.081	0.004	22.327	0.000
FMON15	0.081	0.004	22.327	0.000
FMON16	0.081	0.004	22.327	0.000
FMON17	0.081	0.004	22.327	0.000
FMON18	0.081	0.004	22.327	0.000
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
Residual variances fixed to 0 omitted				
FRISKY12	7.328	0.328	22.349	0.000
FRISKY13	7.328	0.328	22.349	0.000
FRISKY14	7.328	0.328	22.349	0.000
FRISKY15	7.328	0.328	22.349	0.000
FRISKY16	7.328	0.328	22.349	0.000
FRISKY17	7.328	0.328	22.349	0.000
FRISKY18	7.328	0.328	22.349	0.000
RINT	14.315	2.030	7.053	0.000
RSLP	0.453	0.081	5.564	0.000
MINT	0.195	0.023	8.371	0.000
MSLP	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESCOR	**0.350**	**0.034**	**10.441**	**0.000**
INTSTD	**-0.455**	**0.083**	**-5.496**	**0.000**
SLPSTD	**-0.254**	**0.104**	**-2.434**	**0.015**
INDBPINT	0.194	0.251	0.772	0.440
INDBPSLP	-0.006	0.024	-0.239	0.811

Model 2c: Partially Directed Path Multivariate MLM in Mplus: Monitor → Risky for WP residuals within L2 model via placeholder syntax
Also demonstrating how to request BP indirect effects

```
TITLE:  Model 2c: Partially Directed Multivariate Growth Model as MLM
        L1 WP effect as direct path specified in L2 BETWEEN
( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 )

MODEL:  ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar);  ! L1 R: Residual variances (labels)
 Rslp  | risky ON age18;        ! Placeholder for R linear age slope
 Rquad | risky ON agesq;        ! Placeholder for R quadratic age slope
 Mslp  | mon3  ON age18;        ! Placeholder for M linear age slope
 WPres | risky ON mon3;         ! NOW placeholder for L1 WP effect M->R

%BETWEEN%
[risky mon3];                   ! Fixed intercepts
 risky mon3 (Rintvar Mintvar);  ! L2 G: Random intercept variances (labels)
[Rquad Rslp Mslp];              ! Fixed age slopes (as defined earlier)
 Rslp Mslp  (Rslpvar Mslpvar);  ! L2 G: Random linear age slope variances
 Rquad@0;                       ! No quadratic age slope variance
risky Rslp ON att4 (XtoYint XtoYslp);  ! Att-> R int, linear age slope
 mon3 Mslp ON att4 (XtoMint XtoMslp);  ! Att-> M int, linear age slope
 risky WITH Rslp (RIntSlp);     ! R Int-slope covariance (label)
  mon3 WITH Mslp (MIntSlp);     ! M Int-slope covariance (label)

! And now both the intercept -> intercept path and the slope -> slope path
! are contextual BP effects given the L1 placeholder for WP residual effect

 risky ON   mon3  (IntCont);    ! NOW contextual BP intercept effect
 Rslp  ON   Mslp  (SlpCont);    ! NOW contextual BP slope effect
 mon3  WITH Rslp  (Int2Slp);    ! L2 G: M int, R slope covariance
 Mslp  WITH risky (Slp2Int);    ! L2 G: M slope, R int covariance

 [WPres] (ResEff);  ! Fixed effect for L1 WP M->R (as defined earlier)
  WPres@0;          ! No random L1 WP M->R effect variance

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each new created effect -- all values from undirected model
NEW(ResStd BPIntEff IntStd BPSlpEff SlpStd indBPint indBPslp);

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStd = ResEff   * SQRT(0.08077) / SQRT(8.3538);    ! STD WP Res effect
  BPIntEff = ResEff + IntCont;                         ! WP + Context = BP int
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644);   ! STD BP Int effect
  BPSlpEff = ResEff + SlpCont;                         ! WP + Context = BP slp
  SlpStd = BPSlpEff * SQRT(0.01049) / SQRT(0.48830);   ! STD BP Slope effect
  indBPint = XtoMint * BPIntEff;                      ! BP intercept indirect effect
  indBPslp = XtoMslp * BPSlpEff;                      ! BP age slope indirect effect
```

This is still the same model, just with a different syntax for specifying the same level-1 residual to residual directed relationship. This is the version that is necessary in order to have the level-1 effect become random or systematically varying.

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
Residual Variances				
RISKY	7.329	0.328	22.353	0.000
MON3	0.081	0.004	22.354	0.000
Between Level – changed parameters are in BOLD underline				
RSLP ON				
MSLP	-5.294	0.806	-6.569	0.000
RSLP ON				
ATT4	-0.512	0.105	-4.869	0.000
MSLP ON				
ATT4	0.003	0.014	0.240	0.810
RISKY ON				
ATT4	-3.354	0.528	-6.348	0.000
MON3	**-7.938**	**0.872**	**-9.099**	**0.000**
MON3 ON				
ATT4	-0.044	0.057	-0.779	0.436
RISKY WITH				
RSLP	1.480	0.345	4.285	0.000
MSLP	0.039	0.038	1.023	0.306
MON3 WITH				
MSLP	0.000	0.004	-0.105	0.916
RSLP	-0.107	0.031	-3.454	0.001
Means				
RQUAD	0.147	0.021	7.117	0.000
WPRES (here now)	**3.559**	**0.301**	**11.810**	**0.000**
Intercepts				
RISKY	23.598	0.338	69.837	0.000
MON3	0.063	0.034	1.839	0.066
RSLP	1.969	0.139	14.195	0.000
MSLP	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
WPRES	0.000	0.000	999.000	999.000
Residual Variances				
RISKY	14.315	2.030	7.053	0.000
MON3	0.195	0.023	8.371	0.000
RSLP	0.453	0.081	5.564	0.000
MSLP	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESSTD	0.350	0.030	11.810	0.000
BPINTEFF	**-4.380**	**0.797**	**-5.496**	**0.000**
INTSTD	-0.455	0.083	-5.496	0.000
BPSLPEFF	-1.735	0.713	-2.434	0.015
SLPSTD	-0.254	0.104	-2.434	0.015
INDBPINT	**0.194**	**0.251**	**0.772**	**0.440**
INDBPSLP	**-0.006**	**0.024**	**-0.239**	**0.811**

In Mplus, Model 2c as a partially directed single-level SEM: Monitor → Risky for WP residuals using original residuals

```
TITLE:  Model 2c: Partially Directed Multivariate Growth Model as Single-
        Level SEM, L1 WP effect as direct path using original residuals

( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 )
MODEL:  ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0
 risky12-risky18 (Rresvar);        ! L1 R: R residual variances held equal
 mon12-mon18     (Mresvar);        ! L1 R: M residual variances held equal

! Risky behavior quadratic growth model using exact age as loadings
  Rint Rslp Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear growth model using exact age as loadings
  Mint Mslp | mon12-mon18 AT age12-age18;
! Fixed growth effects for R and M
  [Rint Rslp Rquad Mint Mslp];
! L2 G: Random int and linear age slope variances, no quad age variance
  Rint Rslp Mint Mslp (Rintvar Rslpvar Mintvar Mslpvar); Rquad@0;
! L2 G: Within-variable random int-slope covariances for R, M
  Rint WITH Rslp (Rintslp); Mint WITH Mslp (Mintslp);
! Attitudes --> R int, R linear slope, M int, M linear slope
  Rint Rslp Mint Mslp ON att4 (XtoYint XtoYslp XtoMint XtoMslp);

  Rint ON Mint  (IntCont);    ! NOW contextual BP intercept effect
  Rslp ON Mslp  (SlpCont);    ! NOW contextual BP age slope effect
  Mint WITH Rslp (Int2Slp);   ! L2 G: M int, R slope covariance
  Mslp WITH Rint (Slp2Int);   ! L2 G: M slope, R int covariance

! Residual WP covariance between same ages, held equal across age
  risky12-risky18 PON mon12-mon18 (ResEff);

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each created effect -- all values from undirected model
NEW(ResStd BPIntEff IntStd BPSlpEff SlpStd indBPint indBPslp);

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
ResStd = ResEff   * SQRT(0.08077) / SQRT(8.3538);  ! STD WP Res effect
BPIntEff = ResEff + IntCont;                       ! WP + Context = BP int
IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644); ! STD BP Int effect
BPSlpEff = ResEff + SlpCont;                       ! WP + Context = BP slp
SlpStd = BPSlpEff * SQRT(0.01049) / SQRT(0.48830); ! STD BP Slope effect
indBPint = XtoMint * BPIntEff;              ! BP intercept indirect effect
indBPslp = XtoMslp * BPSlpEff;              ! BP age slope indirect effect
```

This is still the same model, just with a different syntax for specifying the same level-1 residual to residual directed relationship. The consequence is that the intercept to intercept and slope to slope relationships become contextual BP effects. Oddly, if we were to switch to ON for the int-slope cross-variable relationships, those stay total BP (see chapter 9 for an example using this version of the model).



For balanced time, a linear growth model would look like this instead (add quad as third place before |):

```
Mint Mslp | mon12@-6 mon13@-5 mon14@-4 mon15@-3
            mon16@-2 mon17@-1 mon18@0;
```

MODEL RESULTS – **changed parameters are in BOLD underline**

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Factor loadings set to 1 omitted				
RINT ON				
MINT	**-7.939**	**0.872**	**-9.099**	**0.000**
RSLP ON				
MSLP	**-5.294**	**0.806**	**-6.569**	**0.000**
RINT ON				
ATT4	-3.354	0.528	-6.348	0.000
RSLP ON				
ATT4	-0.512	0.105	-4.869	0.000
MINT ON				
ATT4	-0.044	0.057	-0.779	0.436
MSLP ON				
ATT4	0.003	0.014	0.240	0.810
RISKY12 ON				
MON12	3.559	0.301	11.810	0.000
RISKY13 ON				
MON13	3.559	0.301	11.810	0.000
RISKY14 ON				
MON14	3.559	0.301	11.810	0.000
RISKY15 ON				
MON15	3.559	0.301	11.810	0.000
RISKY16 ON				
MON16	3.559	0.301	11.810	0.000
RISKY17 ON				
MON17	3.559	0.301	11.810	0.000
RISKY18 ON				
MON18	3.559	0.301	11.810	0.000
RINT WITH				
RSLP	1.480	0.345	4.285	0.000
MSLP	0.039	0.038	1.023	0.306
MINT WITH				
MSLP	0.000	0.004	-0.105	0.916
RSLP	-0.107	0.031	-3.454	0.001

Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
Intercepts fixed to 0 omitted				
RINT	23.598	0.338	69.836	0.000
RSLP	1.969	0.139	14.195	0.000
MINT	0.063	0.034	1.839	0.066
MSLP	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
RISKY12	7.329	0.328	22.353	0.000
RISKY13	7.329	0.328	22.353	0.000
RISKY14	7.329	0.328	22.353	0.000
RISKY15	7.329	0.328	22.353	0.000
RISKY16	7.329	0.328	22.353	0.000
RISKY17	7.329	0.328	22.353	0.000
RISKY18	7.329	0.328	22.353	0.000
MON12	0.081	0.004	22.354	0.000
MON13	0.081	0.004	22.354	0.000
MON14	0.081	0.004	22.354	0.000
MON15	0.081	0.004	22.354	0.000
MON16	0.081	0.004	22.354	0.000
MON17	0.081	0.004	22.354	0.000
MON18	0.081	0.004	22.354	0.000
RINT	14.315	2.030	7.053	0.000
RSLP	0.453	0.081	5.564	0.000
MINT	0.195	0.023	8.371	0.000
MSLP	0.010	0.001	7.802	0.000

New/Additional Parameters

RESSTD	0.350	0.030	11.810	0.000
BPINTEFF	**-4.380**	**0.797**	**-5.496**	**0.000**
INTSTD	-0.455	0.083	-5.496	0.000
BPSLPEFF	**-1.735**	**0.713**	**-2.434**	**0.015**
SLPSTD	-0.254	0.104	-2.434	0.015
INDBPINT	0.194	0.251	0.772	0.440
INDBPSLP	-0.006	0.024	-0.239	0.811

By popular demand, here is an example of how to use "structured residuals" to fit two cross-lag effects at level 1: Model 3a, which switches to covariances at level 2 when fitting these models (per convention, to be agnostic as to which comes first)

```
TITLE:  Model 3a: SEM Structured Residuals to Fit 2 Cross-Lag Paths
 ( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 )

MODEL:  ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0

! Risky behavior quadratic growth model using exact age as loadings
  Rint Rslp Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear growth model using exact age as loadings
  Mint Mslp | mon12-mon18 AT age12-age18;
! Fixed growth effects for R and M
  [Rint Rslp Rquad Mint Mslp];
! L2 G: Random int and linear age slope variances, no quad age variance
  Rint Rslp Mint Mslp (Rintvar Rslpvar Mintvar Mslpvar); Rquad@0;
! Attitudes --> R int, R linear slope, M int, M linear slope
  Rint Rslp Mint Mslp ON att4 (XtoYint XtoYslp XtoMint XtoMslp);
! L2 G: covariances for random intercepts and slopes across outcomes
  Rint Rslp Mint Mslp WITH Rint Rslp Mint Mslp;

! Define new latent factors for residuals at each occasion
  Frisky12 BY risky12@1; Frisky13 BY risky13@1; Frisky14 BY risky14@1;
  Frisky15 BY risky15@1; Frisky16 BY risky16@1; Frisky17 BY risky17@1;
  Frisky18 BY risky18@1; Fmon12 BY mon12@1; Fmon13 BY mon13@1;
  Fmon14 BY mon14@1; Fmon15 BY mon15@1; Fmon16 BY mon16@1;
  Fmon17 BY mon17@1; Fmon18 BY mon18@1;
! All factor means fixed to 0
  [Frisky12-Frisky18@0 Fmon12-Fmon18@0];
! Shut off old residual variances
  risky12-risky18@0 mon12-mon18@0;
! Hold new residual variances equal over time
  Frisky12-Frisky18 (Rresvar); ! L1 R: R residual variances held equal
  Fmon12-Fmon18     (Mresvar); ! L1 R: M residual variances held equal

! Factor residual WP effect between same ages, held equal across age
  Frisky12-Frisky18 PWITH Fmon12-Fmon18 (ResCov);

! Cross-lag WP effects predicting next age, held equal across age
  Frisky13-Frisky18 PON Fmon12-Fmon17     (MR2RR);
  Fmon13-Fmon18     PON Frisky12-Frisky17 (RR2MR);

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each new created effect – using actual values
NEW(ResCor MR2RRsd RR2MRsd);
! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov  / (SQRT(8.3538)*SQRT(0.08077));
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  MR2RRsd = MR2RR * SQRT(0.08077) / SQRT(8.3538);  ! STD M->R lag effect
  RR2MRsd = RR2MR * SQRT(8.3538)  / SQRT(0.08077); ! STD R->M lag effect
```

```
MODEL RESULTS – Parameters fixed to 0 or 1 are omitted for brevity

                                                    Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value
 FRISKY13   ON
    FMON12            -0.255      0.373     -0.682      0.495
 FRISKY14   ON
    FMON13            -0.255      0.373     -0.682      0.495
 FRISKY15   ON
    FMON14            -0.255      0.373     -0.682      0.495
 FRISKY16   ON
    FMON15            -0.255      0.373     -0.682      0.495
 FRISKY17   ON
    FMON16            -0.255      0.373     -0.682      0.495
 FRISKY18   ON
    FMON17            -0.255      0.373     -0.682      0.495

 FMON13     ON
    FRISKY12           0.008      0.004      2.082      0.037
 FMON14     ON
    FRISKY13           0.008      0.004      2.082      0.037
 FMON15     ON
    FRISKY14           0.008      0.004      2.082      0.037
 FMON16     ON
    FRISKY15           0.008      0.004      2.082      0.037
 FMON17     ON
    FRISKY16           0.008      0.004      2.082      0.037
 FMON18     ON
    FRISKY17           0.008      0.004      2.082      0.037

 RINT       ON
    ATT4              -3.156      0.551     -5.725      0.000
 RSLP       ON
    ATT4              -0.516      0.104     -4.945      0.000
 MINT       ON
    ATT4              -0.045      0.057     -0.796      0.426
 MSLP       ON
    ATT4               0.003      0.014      0.221      0.825

RINT     WITH
    RSLP               1.901      0.357      5.318      0.000
    MINT              -0.880      0.170     -5.172      0.000
    MSLP               0.033      0.039      0.847      0.397
 MINT     WITH
    MSLP              -0.001      0.004     -0.186      0.852
    RSLP              -0.110      0.031     -3.529      0.000
 RSLP     WITH
    MSLP              -0.020      0.008     -2.642      0.008
```

```
 FRISKY12 WITH
    FMON12             0.298      0.031      9.607      0.000
 FRISKY13 WITH
    FMON13             0.298      0.031      9.607      0.000
 FRISKY14 WITH
    FMON14             0.298      0.031      9.607      0.000
 FRISKY15 WITH
    FMON15             0.298      0.031      9.607      0.000
 FRISKY16 WITH
    FMON16             0.298      0.031      9.607      0.000
 FRISKY17 WITH
    FMON17             0.298      0.031      9.607      0.000
 FRISKY18 WITH
    FMON18             0.298      0.031      9.607      0.000

Means
    RQUAD              0.146      0.020      7.195      0.000

 Intercepts
    RINT              23.320      0.347     67.184      0.000
    RSLP               1.971      0.136     14.464      0.000
    MINT               0.063      0.034      1.844      0.065
    MSLP              -0.003      0.008     -0.369      0.712

Variances
    FRISKY12           8.301      0.379     21.890      0.000
    FMON12             0.081      0.004     22.125      0.000
    RQUAD              0.000      0.000    999.000    999.000

 Residual Variances
    FRISKY13           8.301      0.379     21.890      0.000
    FRISKY14           8.301      0.379     21.890      0.000
    FRISKY15           8.301      0.379     21.890      0.000
    FRISKY16           8.301      0.379     21.890      0.000
    FRISKY17           8.301      0.379     21.890      0.000
    FRISKY18           8.301      0.379     21.890      0.000

    FMON13             0.081      0.004     22.126      0.000
    FMON14             0.081      0.004     22.126      0.000
    FMON15             0.081      0.004     22.126      0.000
    FMON16             0.081      0.004     22.126      0.000
    FMON17             0.081      0.004     22.126      0.000
    FMON18             0.081      0.004     22.126      0.000

    RINT              18.153      2.208      8.223      0.000
    RSLP               0.492      0.080      6.132      0.000
    MINT               0.194      0.023      8.300      0.000
    MSLP               0.010      0.001      7.676      0.000

New/Additional Parameters
    RESCOR             0.363      0.031     11.624      0.000
    MR2RRSD           -0.025      0.037     -0.681      0.496
    RR2MRSD            0.077      0.037      2.083      0.037
```

What if we controlled for the contemporaneous effect of M → before examining the lagged effect of M → R (my own preference)?

```
TITLE:  Model 3b: Example of Structured Residuals to Fit M->R Cross-Lag Path
        Controls for contemporaneous effect before fitting the lagged effect
```

All else is the same until here…

```
! Factor residual WP effect between same ages, held equal across age
  Frisky12-Frisky18 PON Fmon12-Fmon18 (ResEff);

! Cross-lag WP effects predicting next age, held equal across age
  Frisky13-Frisky18 PON Fmon12-Fmon17 (MR2RR);

MODEL CONSTRAINT:
NEW(ResStD MR2RRsd);
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStD = ResEff * SQRT(Mresvar) / SQRT(Rresvar);  ! STD M->R contemporaneous
  MR2RRsd = MR2RR * SQRT(Mresvar) / SQRT(Rresvar);  ! STD M->R lag effect

                                                   Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value
New/Additional Parameters
    RESSTD             0.335      0.031     10.789      0.000
    MR2RRSD           -0.056      0.034     -1.654      0.098
```

```
TITLE:  Model 3c: Example of Structured Residuals to Fit R->M Cross-Lag Path
        Controls for contemporaneous effect before fitting the lagged effect
```

All else is the same until here…

```
! Factor residual WP effect between same ages, held equal across age
  Fmon12-Fmon18 PON Frisky12-Frisky18 (ResEff);

! Cross-lag WP effects predicting next age, held equal across age
  Fmon13-Fmon18 PON Frisky12-Frisky17 (RR2MR);

MODEL CONSTRAINT:
NEW(ResStD RR2MRsd);
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStD = ResEff * SQRT(8.3538) / SQRT(0.08077); ! STD R->M contemporaneous
  RR2MRsd = RR2MR * SQRT(8.3538) / SQRT(0.08077); ! STD R->M lag effect

                                                   Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value
New/Additional Parameters
    RESSTD             0.373      0.031     12.097      0.000
    RR2MRSD            0.087      0.034      2.570      0.010
```

It looks like evidence for a lagged R → M effect is even stronger after controlling for the contemporaneous effect (and vice-versa).

Here is a comparison of the SEM cross-lagged effects to those from MLM using either manually lagged predictors (in WITHIN only, using ML or BAYES), and to those from MLM using the new LAGGED option (BAYES only). To do so, I had to move all fixed-only L1 effects to the WITHIN model (not use L1 placeholders).

Here is Model 3a using new LAGGED and BAYES: code is same when using my own lagged versions (in which occasion12 = 0), except that lagged variables get added to USEVARIABLE and WITHIN lines, and are used in %WITHIN% model in place of mon3&1 or risky&1 (or ^mon3 or ^risky in Mplus 8.1).

```
DATA:    FILE = Example4Lag.csv;   ! Different data
VARIABLE: ! List of variables in data file
  NAMES = PersonID occasion risky age18 att4 mon3 agesq LagRisky LagMon3 Use12;
! Variables to be analyzed in this model
  USEVARIABLE = age18 agesq att4 risky mon3;
  MISSING ARE ALL (-999);   ! Missing data identifier
! MLM options
  CLUSTER = PersonID;           ! Level-2 ID
  BETWEEN = att4;               ! Observed ONLY level-2 predictors
  WITHIN  = age18 agesq;        ! Observed ONLY level-1 predictors
! Was removed when I used my own lagged variables
  LAGGED  = risky(1) mon3(1);   ! Create Mplus lag-1 versions

ANALYSIS:   TYPE = TWOLEVEL RANDOM; ESTIMATOR = BAYES; ! OR ML
MODEL:  ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar);  ! L1 R: Residual variances (labels)
 Rslp  | risky ON age18;        ! Placeholder for R linear age slope
 risky ON agesq;                ! R fixed quadratic age slope
 Mslp  | mon3  ON age18;        ! Placeholder for M linear age slope
 risky WITH mon3  (ResCov);     ! L1 WP covariance for contemp M->R
 risky ON mon3&1  (MRLagEff);   ! L1 WP fixed effect of lagged M->R
 mon3  ON risky&1 (RMLagEff);   ! L1 WP fixed effect of lagged R->M
! New in 8.1 to create lagged residual effects instead (remove other lag lines)
 risky^ ON mon3^1  (MRLagEff);   ! L1 WP fixed effect of ^res lagged M->R
 mon3^  ON risky^1 (RMLagEff);   ! L1 WP fixed effect of ^res lagged R->M

%BETWEEN%
[risky mon3];                   ! Fixed intercepts
 risky mon3  (Rintvar Mintvar);  ! L2 G: Random intercept variances (labels)
[Rslp Mslp];                    ! Fixed age slopes (as defined earlier)
 Rslp Mslp   (Rslpvar Mslpvar);  ! L2 G: Random linear age slope variances
 risky Rslp ON att4 (XtoYint XtoYslp);  ! Att-> R int, linear age slope
  mon3 Mslp ON att4 (XtoMint XtoMslp);  ! Att-> M int, linear age slope
! L2 G: covariances for random intercepts and slopes across outcomes
  risky Rslp mon3 Mslp WITH risky Rslp mon3 Mslp;

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
NEW(ResCor MR2RRsd RR2MRsd);
! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov / (SQRT(8.3538)*SQRT(0.08077));
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  MR2RRsd = MRLagEff * SQRT(0.08077) / SQRT(8.3538);  ! STD M->R lag effect
  RR2MRsd = RMLagEff * SQRT(8.3538)  / SQRT(0.08077); ! STD R->M lag effect
```

SEs for the lagged effects are noticeably different when using a non-model-provided lagged variable. Tread carefully in this case—it appears to be better to fit cross-lagged effects one of the other two ways.

However: we cannot discern the reason for the differences between the SEM lagged version and the MLM LAGGED version because they must use different estimators (ML vs Bayes; LAGGED is not available in ML and SEM cannot use BAYES with TSCORES).

```
3a SEM via ML and its lagged effects
                                                     Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value
    RESCOR             0.363      0.031     11.624      0.000  *
    MR2RRSD           -0.025      0.037     -0.681      0.496
    RR2MRSD            0.077      0.037      2.083      0.037  *

3a MLM via ML and my lagged effects
                                                     Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value
    RESCOR             0.341      0.034     10.036      0.000  *
    MR2RRSD           -0.051      0.022     -2.310      0.021  *
    RR2MRSD            0.017      0.015      1.115      0.265

3a MLM via Bayes and my lagged effects
                                Posterior         95% C.I.
                    Estimate       S.D.    Lower 2.5%  Upper 2.5%
    RESCOR             0.336       0.033       0.276       0.405  *
    MR2RRSD           -0.054       0.023      -0.092      -0.004  *
    RR2MRSD            0.013       0.016      -0.014       0.045

3a MLM via Bayes and their lagged effects
                                Posterior         95% C.I.
                    Estimate       S.D.    Lower 2.5%  Upper 2.5%
    RESCOR             0.349       0.034       0.288       0.419  *
    MR2RRSD           -0.048       0.025      -0.101      -0.003  *
    RR2MRSD            0.028       0.024      -0.018       0.076
```

The new version of residual lagged effects below appears to match the SEM version most closely:

```
New in 8.1: 3a MLM via Bayes and their ^residual lagged effects
    RESCOR             0.366       0.039       0.300       0.449  *
    MR2RRSD           -0.021       0.035      -0.085       0.050
    RR2MRSD            0.077       0.033       0.012       0.144  *
```

Here are Models 3b and 3c using new LAGGED and BAYES: code is same when using my own lagged versions (in which occasion12 = 0), except that lagged variables get added to USEVARIABLE and WITHIN lines, and are used in %WITHIN% model in place of mon3&1 or risky&1 (or ^mon3 or ^risky in 8.1).

```
DATA:    FILE = Example4Lag.csv;   ! Different data
VARIABLE: ! List of variables in data file
  NAMES = PersonID occasion risky age18 att4 mon3 agesq LagRisky LagMon3 Use12;
! Variables to be analyzed in this model
  USEVARIABLE = age18 agesq att4 risky mon3;
  MISSING ARE ALL (-999);   ! Missing data identifier
! MLM options
  CLUSTER = PersonID;            ! Level-2 ID
  BETWEEN = att4;                ! Observed ONLY level-2 predictors
  WITHIN  = age18 agesq;         ! Observed ONLY level-1 predictors
! LAGGED was removed when I used my own lagged variables
  LAGGED  = mon3(1);      ! For Model 3b: Create Mplus lag-1 version
  LAGGED  = risky(1);     ! For Model 3c: Create Mplus lag-1 version
  LAGGED  = mon3(1) risky(1);    ! For 8.1 res lag-1 versions

ANALYSIS:   TYPE = TWOLEVEL RANDOM; ESTIMATOR = BAYES; ! OR ML
MODEL:  ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar);   ! L1 R: Residual variances (labels)
 Rslp  | risky ON age18;         ! Placeholder for R linear age slope
 risky ON agesq;                 ! R fixed quadratic age slope
 Mslp  | mon3  ON age18;         ! Placeholder for M linear age slope
 risky ON mon3     (ResEff);     ! Model 3b: L1 WP covariance for contemp M->R
 mon3  ON risky    (ResEff);     ! Model 3c: L1 WP covariance for contemp R->M
 risky ON mon3&1   (MRLagEff);   ! Model 3b: L1 WP fixed effect of lagged M->R
 mon3  ON risky&1  (RMLagEff);   ! Model 3c: L1 WP fixed effect of lagged R->M
! New in 8.1 to create lagged residual effects instead (remove other lag lines)
 risky^ ON mon3^1  (MRLagEff);   ! Model 3b: L1 WP fixed eff of res lagged M->R
 mon3^  ON risky^1 (RMLagEff);   ! Model 3c: L1 WP fixed eff of res lagged R->M

%BETWEEN%
[risky mon3];                    ! Fixed intercepts
 risky mon3  (Rintvar Mintvar);  ! L2 G: Random intercept variances (labels)
[Rslp Mslp];                     ! Fixed age slopes (as defined earlier)
 Rslp Mslp   (Rslpvar Mslpvar);  ! L2 G: Random linear age slope variances
 risky Rslp ON att4 (XtoYint XtoYslp);  ! Att-> R int, linear age slope
  mon3 Mslp ON att4 (XtoMint XtoMslp);  ! Att-> M int, linear age slope
! L2 G: covariances for random intercepts and slopes across outcomes
  risky Rslp mon3 Mslp WITH risky Rslp mon3 Mslp;

MODEL CONSTRAINT:  ! Like ESTIMATE in SAS, but can refer to any parameter
! Need to name each new created effect -- using actual values
NEW(ResStd MR2RRsd RR2MRsd);
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStD  = ResEff   * SQRT(0.08077) / SQRT(8.3538);  ! 3b: STD M->R contemp
  MR2RRsd = MRLagEff * SQRT(0.08077) / SQRT(8.3538);  ! 3b: STD M->R lag effect
  ResStD  = ResEff   * SQRT(8.3538)  / SQRT(0.08077); ! 3c: STD R->M contemp
  RR2MRsd = RMLagEff * SQRT(8.3538)  / SQRT(0.08077); ! 3c: STD R->M lag effect
```

3b SEM via ML and its lagged effects

	Estimate	S.E.	Est./S.E.	2-T P-Value	
RESSTD	0.335	0.031	10.789	0.000	*
MR2RRSD	-0.056	0.034	-1.654	0.098	

3b MLM via ML and my lagged effects

	Estimate	S.E.	Est./S.E.	2-T P-Value	
RESSTD	0.336	0.030	11.185	0.000	*
MR2RRSD	-0.053	<u>0.022</u>	-2.438	0.015	*

3b MLM via Bayes and my lagged effects

	Estimate	Posterior S.D.	95% C.I. Lower 2.5%	Upper 2.5%	
RESSTD	0.341	0.027	0.288	0.391	*
MR2RRSD	-0.047	<u>0.022</u>	-0.092	-0.006	*

3b MLM via Bayes and their lagged effects

	Estimate	Posterior S.D.	95% C.I. Lower 2.5%	Upper 2.5%	
RESSTD	0.330	0.032	0.265	0.391	*
MR2RRSD	-0.059	<u>0.022</u>	-0.097	-0.013	*

<u>New in 8.1: 3b MLM via Bayes and their ^residual lagged effects</u>

	Estimate	**Posterior S.D.**	**95% C.I. Lower 2.5%**	**Upper 2.5%**	
RESSTD	**0.330**	**0.028**	**0.276**	**0.377**	*
MR2RRSD	**-0.058**	**<u>0.033</u>**	**-0.123**	**0.005**	

3c SEM via ML and its lagged effects

	Estimate	S.E.	Est./S.E.	2-T P-Value	
RESSTD	0.373	0.031	12.097	0.000	*
RR2MRSD	0.087	0.034	2.570	0.010	*

3c MLM via ML and my lagged effects

	Estimate	S.E.	Est./S.E.	2-T P-Value	
RESSTD	0.353	0.030	11.858	0.000	*
RR2MRSD	0.041	<u>0.015</u>	2.755	0.006	*

3c MLM via Bayes and my lagged effects

	Estimate	Posterior S.D.	95% C.I. Lower 2.5%	Upper 2.5%	
RESCOR	0.353	0.027	0.301	0.404	*
RR2MRSD	0.045	<u>0.014</u>	0.017	0.069	*

3c MLM via Bayes and their lagged effects

	Estimate	Posterior S.D.	95% C.I. Lower 2.5%	Upper 2.5%	
RESCOR	0.327	0.027	0.271	0.381	*
RR2MRSD	0.005	<u>0.023</u>	-0.038	0.051	

<u>New in 8.1: 3c MLM via Bayes and their ^residual lagged effects</u>

	Estimate	**Posterior S.D.**	**95% C.I. Lower 2.5%**	**Upper 2.5%**	
RESCOR	**0.361**	**0.031**	**0.289**	**0.414**	*
RR2MRSD	**0.084**	**<u>0.036</u>**	**0.013**	**0.155**	*